
Solution



History of Neo4j

• 0.x ... 
small embeddable persistent graph library

• 1.x ... 
adding indexes, server, first stab of Cypher

• 2.x ... 
ease of use, data-model, optional schema, 
cost based optimizer, import, Neo4j-Browser

• 3.x …
binary protocol, bytecode compiled queries,
sharding

Value from Data Relationships
Common Use Cases

Internal Applications

Master Data Management 

Network and 
IT Operations

Fraud Detection

Customer-Facing Applications

Real-Time Recommendations

Graph-Based Search

Identity and 
Access Management

http://neo4j.com/use-cases



The Whiteboard Model Is the Physical Model



CAR

DRIVES

name: "Dan"
born: May 29, 1970

twitter: "@dan"
name: "Ann"

born:  Dec 5, 1975

since: 
Jan 10, 2011

brand: "Volvo"
model: "V70"

Property Graph Model

Nodes
• The objects in the graph
• Can have name-value properties
• Can be labeled

Relationships
• Relate nodes by type and direction
• Can have name-value properties

LOVES

LOVES

LIVES WITH

NS

PERSON PERSON

http://neo4j.com/developer/graph-database/#property-graph



Relational to Graph

Relational Graph

http://neo4j.com/developer/graph-db-vs-rdbms/



Neo4j: All About Patterns

               (:Person { name:"Dan"} ) -[:LOVES]-> (:Person { name:"Ann"} ) 

LOVES

Dan Ann

LABEL PROPERTY

NODE NODE

LABEL PROPERTY

http://neo4j.com/developer/cypher



Cypher: Find Patterns

      MATCH (:Person { name:"Dan"} ) -[:LOVES]-> (love:Person) RETURN love

LOVES

Dan ?

LABEL

NODE NODE

LABEL PROPERTY ALIAS ALIAS

http://neo4j.com/developer/cypher



Introducing our data set...



meetup.com’s recommendations



Recommendation queries

‣ Several different types
• groups to join
• topics to follow
• events to attend

‣ As a user of meetup.com trying to find 
groups to join, events to attend and people 
to meet



How will this talk be structured?





Data ?

‣ Groups
‣ Members
‣ Events
‣ Topics
‣ Time & Date
‣ Location



Get Data: Meetup API + jq

stedolan.github.io/jq/meetup.com/meetup_api/



Find similar groups to Neo4j

As a member of the Neo4j London group

I want to find other similar meetup groups

So that I can join those groups



What makes groups similar?



As a member of the Neo4j London group

I want to find other similar meetup groups

So that I can join those groups

Find similar groups to Neo4j



LOAD CSV FROM "file:///groups.csv" 

AS row

RETURN row LIMIT 5;

LOAD CSV WITH HEADERS FROM "file:///groups.csv" 

AS row WITH row

WHERE row.rating > 4.5

RETURN row;

LOAD CSV



+-----------+------------------------------+----------------------------+--------+----------------

+|  id       | name                         | urlname                    | rating | created        

|

|-----------+------------------------------+----------------------------+--------+----------------|

|  841735   | LJC - London Java Community  | Londonjavacommunity        | 4.54   | 1196081014000  |

|  18313232 | Kubernetes London            | Kubernetes-London          | 5      | 1420729836000  |

|  18581527 | data+visual London           | data-visual-London         | 4.67   | 1431021679000  |

|  163876   | London Web                   | londonweb                  | 4.11   | 1034097743000  |

|  15734842 | Ansible London               | Ansible-London             | 4.42   | 1405439359000  |

|  12963902 | Scalability London           | Scalability-London         | 4.95   | 1392824462000  |

|  4062902  | Ember London                 | London-Emberjs-User-Group  | 4.66   | 1339522219000  |

+-----------+------------------------------+----------------------------+--------+----------------+

groups.csv



LOAD CSV WITH HEADERS FROM "file:///groups.csv" 

AS row

CREATE (:Group { id:row.id,

                 name:row.name,

                 urlname:row.urlname,

                 rating:toInt(row.rating),

                 created:toInt(row.created) })

Create groups



LOAD CSV WITH HEADERS FROM "file:///groups.csv" 

AS row

CREATE (:Group { id:row.id,

                 name:row.name,

                 urlname:row.urlname,

                 rating:toint(row.rating),

                 created:toint(row.created) })

Create groups

We use CREATE because the database is empty.



groups_topics.csv

|----------+---------------------------+--------------------------|

|  id      | name                      | urlkey                   |

|----------+---------------------------|--------------------------|
|  827     | .NET                      | dotnet                   |
|  2109    | System Administration     | sysadmin                 |
|  2260    | C#                        | csharp                   |
|  10105   | Microsoft Windows         | mswindows                |
|  15167   | Cloud Computing           | cloud-computing          |
|  46810   | Configuration Management  | configuration-management |
|  52210   | PowerShell                | powershell               |
|  66339   | Windows Azure Platform    | windows-azure-platform   |
|  84706   | Scripting                 | scripting                |
|  87614   | DevOps                    | devops                   |
|  99537   | Microsoft Technology      | microsoft-technology     |
|  189     | Java                      | java                     |
|  563     | Open Source               | opensource               |
|----------+---------------------------+--------------------------|



LOAD CSV WITH HEADERS FROM "file:///groups_topics.csv" 

AS row

MERGE (topic:Topic {id: row.id})

ON CREATE SET topic.name = row.name,

              topic.urlkey = row.urlkey

Create topics



LOAD CSV WITH HEADERS FROM "file:///groups_topics.csv" 

AS row

MERGE (topic:Topic {id: row.id})

ON CREATE SET topic.name = row.name,

              topic.urlkey = row.urlkey

Create topics

We use MERGE because we want to avoid 
creating duplicate topics



CREATE CONSTRAINT ON (t:Topic)

ASSERT t.id IS UNIQUE

CREATE CONSTRAINT ON (g:Group)

ASSERT g.id IS UNIQUE

Create unique constraints



CREATE CONSTRAINT ON (t:Topic)

ASSERT t.id IS UNIQUE

CREATE CONSTRAINT ON (g:Group)

ASSERT g.id IS UNIQUE

Create unique constraints

We create unique constraints to: 
● ensure uniqueness across a (label,property) pair
● allow fast lookup of nodes which match these 

(label,property) pairs.



How does Neo4j use indexes?

Indexes are only used to find the starting 
point for queries. 

Use index scans to look up rows in tables 
and join them with rows from other tables

Use indexes to find the starting 
points for a query. 

Relational Graph



|----------+-----------|

|  id      | groupId   |

|----------+-----------|

|  827     | 18780165  |

|  2109    | 18780165  |

|  2260    | 18780165  |

|  10105   | 18780165  |

|  15167   | 18780165  |

|  46810   | 18780165  |

|  52210   | 18780165  |

|----------+-----------|

Groups and topics



LOAD CSV WITH HEADERS FROM "file:///groups_topics.csv" 

AS row

MATCH (topic:Topic {id: row.id})

MATCH (group:Group {id: row.groupId})

MERGE (group)-[:HAS_TOPIC]->(topic)

Connect groups and topics



LOAD CSV WITH HEADERS FROM "file:///groups_topics.csv" 

AS row

MATCH (topic:Topic {id: row.id})

MATCH (group:Group {id: row.groupId})

MERGE (group)-[:HAS_TOPIC]->(topic)

Connect groups and topics

We can use MERGE to uniquely create 
relationships as well



CREATE INDEX ON :Group(name)

Create index



CREATE INDEX ON :Group(name)

Create index

We create an index on :Group(name) so that 
we can quickly look up groups by name.



Find similar groups to Neo4j

MATCH (group:Group {name: "Neo4j - London User Group"})
      -[:HAS_TOPIC]->(topic)<-[:HAS_TOPIC]-(otherGroup)
RETURN otherGroup.name, 
       COUNT(topic) AS topicsInCommon, 
       COLLECT(topic.name) AS topics
ORDER BY topicsInCommon DESC, otherGroup.name
LIMIT 10



Find similar groups to Neo4j



I’m already a member of these!



Exclude groups I’m a member of

As a member of the Neo4j London group

I want to find other similar meetup groups
that I’m not already a member of

So that I can join those groups



What other data can we get? 



Exclude groups I’m a member of

As a member of the Neo4j London group

I want to find other similar meetup groups
that I’m not already a member of

So that I can join those groups



|------------+--------------------+---------------|

|  id        | name               | joined        |

|------------+--------------------+---------------|

|  103929052 | A                  | 1378461129000 |

|  11337881  | Abhishek Shivkumar | 1421419313000 |

|  39676622  | Ali Syed           | 1395723669000 |

|  2773509   | Amit               | 1407935487000 |

|  30225872  | Attila Sztupak     | 1378812292000 |

|  12882650  | Cathy White        | 1423566263000 |

|  109548702 | Danny Bickson      | 1378196635000 |

|------------+--------------------+---------------|

members.csv



Create members

LOAD CSV WITH HEADERS FROM "file:///path/to/members.csv" AS row

WITH DISTINCT row.id AS id, row.name AS name

MERGE (member:Member {id: id})

ON CREATE SET member.name = name



|------------+-----------|

|  id        | groupId   |

|------------+-----------|

|  103929052 | 10087112  |

|  11337881  | 10087112  |

|  39676622  | 10087112  |

|  2773509   | 10087112  |

|  30225872  | 10087112  |

|  12882650  | 10087112  |

|  109548702 | 10087112  |

|------------+-----------|

Members and groups



LOAD CSV WITH HEADERS FROM "file:///path/to/members.csv" AS row

WITH row WHERE NOT row.joined is null

MATCH (member:Member {id: row.id})

MATCH (group:Group {id: row.groupId})

MERGE (member)-[:MEMBER_OF {joined: toint(row.joined)}]->(group)

Connect members and groups



Exclude groups I’m a member of

MATCH (group:Group {name: "Neo4j - London User Group"})       
       -[:HAS_TOPIC]->(topic)<-[:HAS_TOPIC]-(otherGroup:Group)
RETURN otherGroup.name, 
       COUNT(topic) AS topicsInCommon, 
       EXISTS((:Member {name: "Mark Needham"})
               -[:MEMBER_OF]->(otherGroup)) AS alreadyMember,
       COLLECT(topic.name) AS topics
ORDER BY topicsInCommon DESC
LIMIT 10



Exclude groups I’m a member of



Exclude groups I’m a member of

MATCH (group:Group {name: "Neo4j - London User Group"})       
       -[:HAS_TOPIC]->(topic)<-[:HAS_TOPIC]-(otherGroup:Group)
WHERE NOT( (:Member {name: "Mark Needham"})    
            -[:MEMBER_OF]->(otherGroup) )
RETURN otherGroup.name, 
       COUNT(topic) AS topicsInCommon, 
       COLLECT(topic.name) AS topics
ORDER BY topicsInCommon DESC
LIMIT 10



Exclude groups I’m a member of



Find my similar groups

As a member of several meetup groups

I want to find other similar meetup groups       

that I’m not already a member of

So that I can join those groups



Find my similar groups

As a member of several meetup groups

I want to find other similar meetup groups       

that I’m not already a member of

So that I can join those groups



|------------+----------------------------------------------|

|  id        | topics                                       |

|------------+----------------------------------------------|

|  103929052 | 18062;563;16575;20923;3833;108403;1307;10099 |

|  11337881  | 1372;1512;49585;24553;417;24778;25584;23005  |

|  39676622  |                                              |

|  2773509   |                                              |

|  30225872  | 48471;22792;58162;1762                       |

|  12882650  | 563;3833;9696;659;1621,48471;22792           |

|  109548702 | 21681;30928;18062;5532,55324;15167;108403    |

|------------+----------------------------------------------|

Members and topics



USING PERIODIC COMMIT 10000

LOAD CSV WITH HEADERS FROM "file:///path/to/members.csv" AS row

WITH split(row.topics, ";") AS topics, row.id AS memberId

UNWIND topics AS topicId

MATCH (member:Member {id: memberId})

MATCH (topic:Topic {id: topicId})

MERGE (member)-[:INTERESTED_IN]->(topic)

Connect members and topics



Find my similar groups

MATCH (member:Member {name: "Mark Needham"})
      -[:INTERESTED_IN]->(topic), 
      (member)-[:MEMBER_OF]->(group)-[:HAS_TOPIC]->(topic)

WITH member, topic, COUNT(*) AS score
MATCH (topic)<-[:HAS_TOPIC]-(otherGroup) 
WHERE NOT (member)-[:MEMBER_OF]->(otherGroup)
RETURN otherGroup.name, 
       COLLECT(topic.name), 
       SUM(score) as score
ORDER BY score DESC



Find my similar groups



Interests



What am I actually interested in?

There’s an implicit INTERESTED_IN relationship 
between the topics of groups I belong to but 
don’t express an interest in. Let’s make it explicit



What am I actually interested in?

There’s an implicit INTERESTED_IN relationship 
between the topics of groups I belong to but 
don’t express an interest in. Let’s make it explicit

P

HAS_TOPIC

INTERESTED_IN



What am I actually interested in?

MATCH (m:Member)-[:RSVPD {response:"yes"}]->(event)
       <-[:HOSTED_EVENT]->()-[:HAS_TOPIC]->(topic)

WITH m, topic, 
     COUNT(*) AS times

WHERE times > 5 

RETURN m.name, topic.name, times
ORDER BY times DESC



What am I actually interested in?

MATCH (m:Member)-[:RSVPD {response:"yes"}]->(event)
       <-[:HOSTED_EVENT]->()-[:HAS_TOPIC]->(topic)

WITH m, topic, 
     COUNT(*) AS times, 
     COLLECT(event.name) AS events 

WHERE times > 5 

AND NOT (m)-[:INTERESTED_IN]->(topic)
MERGE (m)-[:INTERESTED_IN]->(topic)



What am I actually interested in?



Finally, Events!



Now - let’s recommend events!



Events in my groups

As a member of several meetup groups

I want to find other events hosted by 

those groups

So that I can attend those events



Events in my groups

As a member of several 
meetup groups 

I want to find other events 

hosted by those groups

So that I can attend those 
events



Events

|---------------+---------------------------------------------+---------------+-------------|
|  id           | name                                        | time          | utc_offset  |
|---------------+---------------------------------------------+---------------+-------------|
|  3261890      | London Web Design October Meetup            | 1097776800000 | 3600000     |
|  3492560      | London Web Design November Meetup           | 1100199600000 | 0           |
|  3683911      | London Web Design December Meetup           | 1102618800000 | 0           |
|  4339054      | The London Web Design March Meetup          | 1113413400000 | 3600000     |
|  4825171      | The London PHP January Meetup               | 1136487600000 | 0           |
|  4795898      | January Meetup                              | 1137006000000 | 0           |
|  4826924      | The London PHP February Meetup              | 1138906800000 | 0           |
|  4832622      | The London Web Design February Meetup       | 1140030000000 | 0           |
|  8646860      | JAVAWUG BOF 40 JQuantLib                    | 1221672600000 | 3600000     |
|  8689280      | PHP London October Meetup                   | 1222972200000 | 3600000     |
|  8730923      | The London Cloud Computing October Meetu    | 1223488800000 | 3600000     |
|  8879609      | JWUG BOF41 Web Applications and RESTful     | 1224523800000 | 3600000     |
|  8921257      | OSGi for the Web Developer followed by f    | 1225217700000 | 0           |
|---------------+---------------------------------------------+---------------+-------------|



CREATE INDEX ON :Event(id)

CREATE INDEX ON :Event(time)

LOAD CSV WITH HEADERS FROM "file:///events.csv" AS row

MERGE (event:Event {id: row.id})

ON CREATE SET event.name = row.name,

              event.time = toint(row.time),

              event.utcOffset = toint(row.utc_offset)

Create events



Events and groups

|---------------+-----------|
|  id           | group_id  |
|---------------+-----------|
|  3261890      | 163876    |
|  3492560      | 163876    |
|  3683911      | 163876    |
|  3857967      | 163876    |
|  4339054      | 163876    |
|  4572794      | 163876    |
|  4709866      | 163876    |
|  4772985      | 163876    |
|  4785678      | 163876    |
|  4825171      | 218194    |
|  4826924      | 218194    |
|  4832622      | 163876    |
|  4846072      | 218194    |
|---------------+-----------|



Connect events and groups

LOAD CSV WITH HEADERS FROM "file:///events.csv" AS row

MATCH (group:Group {id: row.group_id})

MATCH (event:Event {id: row.id})

MERGE (group)-[:HOSTED_EVENT]->(event)



WITH 24.0*60*60*1000 AS oneDay

MATCH (member:Member {name: "Mark Needham"}),

      (member)-[:MEMBER_OF]->(group),

      (group)-[:HOSTED_EVENT]->(futureEvent)

WHERE futureEvent.time >= timestamp()

RETURN group.name, futureEvent.name, 

       round((futureEvent.time - timestamp()) / oneDay) AS days

ORDER BY days

LIMIT 10

Events in my groups



Events in my groups



Events in my groups



Events in my groups



Layered recommendations

We can improve our recommendation by 
weighting different attributes:

‣ events in my groups
‣ events I’ve previously attended
‣ topics I’m interested in
‣ events my peers attend



Events in my groups

We can improve our recommendation by 
weighting different attributes:

‣ events in my groups
‣ events I’ve previously attended
‣ topics I’m interested in
‣ events my peers attend



WITH 24.0*60*60*1000 AS oneDay

MATCH (member:Member {name: "Mark Needham"})

MATCH (futureEvent:Event) WHERE futureEvent.time >= timestamp()

MATCH (futureEvent)<-[:HOSTED_EVENT]-(group)

RETURN group.name, 

       futureEvent.name, 

       EXISTS((group)<-[:MEMBER_OF]-(member)) AS isMember,

       round((futureEvent.time - timestamp()) / oneDay) AS days

ORDER BY isMember DESC, days

Events in my groups



Events in my groups



+ previous events attended

We can improve our recommendation by 
weighting different attributes:

‣ events in my groups
‣ events I’ve previously attended
‣ topics I’m interested in
‣ events my peers attend



+ previous events attended

As a member of several meetup 
groups who has previously 
attended events

I want to find other events 

hosted by those groups

So that I can attend those events



RSVPs

|------------+-----------+-----------+--------+----------+---------------+----------------|
|  rsvp_id   | event_id  | member_id | guests | response | created       | mtime          |
|------------+-----------+-----------+--------+----------+---------------+----------------|
|  654924042 | 100056812 | 65110402  | 0      | yes      | 1358436329000 | 1358436329000  |
|  666200862 | 100056812 | 32158012  | 0      | yes      | 1359212092000 | 1359212092000  |
|  655045942 | 100056812 | 45574682  | 0      | yes      | 1358442847000 | 1358442847000  |
|  654946622 | 100056812 | 64073592  | 0      | yes      | 1358437486000 | 1358437486000  |
|  696456002 | 100056812 | 70201982  | 0      | yes      | 1361279846000 | 1361279846000  |
|  689115982 | 100056812 | 12434405  | 0      | yes      | 1360748670000 | 1360748670000  |
|  654924112 | 100056812 | 34168592  | 0      | no       | 1358436332000 | 1358436332000  |
|  654925662 | 100056812 | 3401490   | 0      | no       | 1358436413000 | 1360361799000  |
|  656439652 | 100056812 | 12252389  | 0      | no       | 1358533048000 | 1361197297000  |
|  689112692 | 100056812 | 76908802  | 0      | yes      | 1360748069000 | 1360748069000  |
|  690924922 | 100056812 | 10704191  | 0      | yes      | 1360876122000 | 1360876122000  |
|  690834812 | 100056812 | 71296302  | 0      | yes      | 1360871204000 | 1360871204000  |
|  691120252 | 100056812 | 71730512  | 0      | yes      | 1360888294000 | 1360888294000  |
|------------+-----------+-----------+--------+----------+---------------+----------------|



LOAD CSV WITH HEADERS FROM "file:///rsvps.csv" AS row

MATCH (member:Member {id: row.member_id})

MATCH (event:Event {id: row.event_id})

MERGE (member)-[rsvp:RSVPD {id: row.rsvp_id}]->(event)

ON CREATE SET rsvp.created = toint(row.created), 

              rsvp.lastModified = toint(row.mtime), 

              rsvp.response = row.response;

Create RSVPs



+ previous events attended
WITH 24.0*60*60*1000 AS oneDay

MATCH (member:Member {name: "Mark Needham"})

MATCH (futureEvent:Event) WHERE futureEvent.time >= timestamp()

MATCH (futureEvent)<-[:HOSTED_EVENT]-(group)

WITH oneDay, group, futureEvent, member, EXISTS((group)<-[:MEMBER_OF]-(member)) AS isMember

OPTIONAL MATCH (member)-[rsvp:RSVPD {response: "yes"}]->(pastEvent)<-[:HOSTED_EVENT]-(group)

WHERE pastEvent.time < timestamp()

RETURN group.name, 

       futureEvent.name, 

       isMember, 

       COUNT(rsvp) AS previousEvents,

       round((futureEvent.time - timestamp()) / oneDay) AS days

ORDER BY days, previousEvents DESC



+ previous events attended



RSVP_YES vs RSVPD

I was curious whether refactoring 
RSVPD {response: "yes"} to RSVP_YES would have 
any impact as Neo4j is optimised for querying 
by unique relationship types.



RSVP_YES vs RSVPD

MATCH (m:Member)-[rsvp:RSVPD {response:"yes"}]->(event)

MERGE (m)-[rsvpYes:RSVP_YES {id: rsvp.id}]->(event)

ON CREATE SET rsvpYes.created = rsvp.created,     

              rsvpYes.lastModified = rsvp.lastModified;

MATCH (m:Member)-[rsvp:RSVPD {response:"no"}]->(event)

MERGE (m)-[rsvpYes:RSVP_NO {id: rsvp.id}]->(event)

ON CREATE SET rsvpYes.created = rsvp.created, 

              rsvpYes.lastModified = rsvp.lastModified;



RSVP_YES vs RSVPD

RSVPD {response: "yes"}

vs

RSVP_YES

Cypher version: CYPHER 2.3, 
planner: COST. 
688635 total db hits in 232 ms.

Cypher version: CYPHER 2.3, 
planner: COST. 
559866 total db hits in 207 ms.



+ my topics

We can improve our recommendation by 
weighting different attributes:

‣ events in my groups
‣ events I’ve previously attended
‣ topics I’m interested in
‣ events my peers attend



+ my topics
WITH 24.0*60*60*1000 AS oneDay

MATCH (member:Member {name: "Mark Needham"})

MATCH (futureEvent:Event) WHERE futureEvent.time >= timestamp()

MATCH (futureEvent)<-[:HOSTED_EVENT]-(group)

WITH oneDay, group, futureEvent, member, EXISTS((group)<-[:MEMBER_OF]-(member)) AS isMember

OPTIONAL MATCH (member)-[rsvp:RSVPD {response: "yes"}]->(pastEvent)<-[:HOSTED_EVENT]-(group)

WHERE pastEvent.time < timestamp()

WITH oneDay, group, futureEvent, member, isMember, COUNT(rsvp) AS previousEvents

OPTIONAL MATCH (futureEvent)<-[:HOSTED_EVENT]-()-[:HAS_TOPIC]->(topic)<-[:INTERESTED_IN]-(member)

RETURN group.name, futureEvent.name, isMember, previousEvents, 

       COUNT(topic) AS topics, round((futureEvent.time - timestamp()) / oneDay) AS days

ORDER BY days,previousEvents DESC, topics DESC



+ my topics



+ events my friends are attending

We can improve our recommendation by 
weighting different attributes:

‣ events in my groups
‣ events I’ve previously attended
‣ topics I’m interested in
‣ events my peers attend



+ events my friends are attending

There’s an implicit FRIENDS relationship 
between people who attended the same events.
Let’s make it explicit.



+ events my friends are attending

There’s an implicit FRIENDS relationship 
between people who attended the same events.
Let’s make it explicit.

M

RSVPD



+ events my friends are attending

MATCH (m1:Member)
WHERE NOT m1:Processed

WITH m1 LIMIT {limit}
MATCH (m1)-[:RSVP_YES]->(event:Event)<-[:RSVP_YES]-(m2:Member)

WITH m1, m2, COLLECT(event) AS events, COUNT(*) AS times
WHERE times >= 5

WITH m1, m2, times, [event IN events | SIZE((event)<-[:RSVP_YES]-())] AS attendances

WITH m1, m2, REDUCE(score = 0.0, a IN attendances | score + (1.0 / a)) AS score

RETURN ID(m1) AS m1, ID(m2) AS m2, score



+ events my friends are attending

UNWIND {rows} AS row

MATCH (m1), (m2)
WHERE ID(m1) = row.m1 AND ID(m2) = row.m2

MERGE (m1)-[friendsRel:FRIENDS]-(m2)
SET friendsRel.score = row.score

SET m1:Processed

rows
[ 
  ...
  { 
    "m1": 12345, 
    "m2": 678912, 
    "score": 0.23471 
  },
  ...
]



Bidirectional relationships

‣ You may have noticed that we didn’t specify a 
direction when creating the relationship
MERGE (m1)-[:FRIENDS]-(m2)

‣ FRIENDS is a bidirectional relationship. We only 
need to create it once between two people.

‣ We ignore the direction when querying



+ events my friends are attending
WITH 24.0*60*60*1000 AS oneDay
MATCH (member:Member {name: "Mark Needham"})
MATCH (futureEvent:Event) WHERE futureEvent.time >= timestamp()
MATCH (futureEvent)<-[:HOSTED_EVENT]-(group)

WITH oneDay, group, futureEvent, member, EXISTS((group)<-[:MEMBER_OF]-(member)) AS isMember
OPTIONAL MATCH (member)-[rsvp:RSVPD {response: "yes"}]->(pastEvent)<-[:HOSTED_EVENT]-(group)
WHERE pastEvent.time < timestamp()

WITH oneDay, group, futureEvent, member, isMember, COUNT(rsvp) AS previousEvents
OPTIONAL MATCH (futureEvent)<-[:HOSTED_EVENT]-()-[:HAS_TOPIC]->(topic)<-[:INTERESTED_IN]-(member)

WITH oneDay, group, futureEvent, member, isMember, previousEvents, COUNT(topic) AS topics
OPTIONAL MATCH (member)-[:FRIENDS]-(:Member)-[rsvpYes:RSVP_YES]->(futureEvent)

RETURN group.name, futureEvent.name, isMember, round((futureEvent.time - timestamp()) / oneDay) AS days, 
       previousEvents, topics, COUNT(rsvpYes) AS friendsGoing
ORDER BY days, friendsGoing DESC, previousEvents DESC
LIMIT 15



+ events my friends are attending



Scoring

We’re using a simple count based scoring 
ordering. 

In a production system we might apply 
something more sophisticated e.g. log or 
Pareto function



Real time recommendations



Real time recommendations
{
  "venue": {
    "venue_id": 14544952
  },
  "response": "no",
  "guests": 0,
  "member": { "member_id": 54585732 },
  "rsvp_id": 1579878700,
  "mtime": 1448705224460,
  "event": {
    "event_id": "226676071",
  },
  "group": {
    "group_id": 8501832,
  }
}



Real time recommendations

import requests
import json

def stream_meetup():
    r = requests.get('http://stream.meetup.com/2/rsvps', stream=True)
    for raw_rsvp in r.iter_lines():
        if raw_rsvp:
            yield raw_rsvp



Real time recommendations

from py2neo import authenticate, Graph            

authenticate("localhost:7474", "neo4j", "test")
graph = Graph()

group_ids = []
group_query = "MATCH (g:Group) RETURN g.id AS groupId"
for row in graph.cypher.execute(group_query):
    group_ids.append(int(row["groupId"]))



Real time recommendations
for rsvp in stream_meetup():
    if rsvp["group"]["group_id"] in group_ids:
        params = { "rsvp_id": str(rsvp["rsvp_id"]),
                   "event_id": str(rsvp["event"]["event_id"]),
                   "member_id": str(rsvp["member"]["member_id"]),
                   "response": rsvp["response"],
                   "mtime": rsvp["mtime"] }
        graph.cypher.execute("""
          MATCH (event:Event {id: {event_id}})
          MATCH (member:Member {id: {member_id}})
          MERGE (member)-[rsvpRel:RSVPD {id: {rsvp_id}}]->(event)
          ON CREATE SET rsvpRel.created = toint({mtime})
          ON MATCH  SET rsvpRel.lastModified = toint({mtime})
          SET rsvpRel.response = {response}""", params)



What could we do next?

‣ Comments sentiment analysis
• do people actually like the events they go to?

‣ Topic ontology
• how are topics related? e.g. Neo4j, Cassandra, 

MongoDB are part of NoSQL

‣ Event similarity based on descriptions
• use automated topic derivation to derive categories



What could we do next?

‣ Social network
• what events do our twitter/Facebook friends attend?

‣ Location
• do we favour events in a certain part of town?

‣ Day of the week
• do we only go to events on certain days of the week?

• do we go to different events on weekdays vs 

weekend?



Why Neo4j for recommendations?

‣ Real time
• take into account what you’ve just done

‣ Flexibility
• bring information from different sources and evolve data model as 

needed for use-cases
• easily combine collaborative + content filtering in a single query

‣ Intuitive query language
• focus on describing the domain problem. Even non technical users 

can read our queries.



That’s all for today!
Questions? :-)

Michael Hunger @mesirii
created by

Mark Needham @markhneedham
https://github.com/neo4j-meetups/modeling-worked-example

https://github.com/neo4j-meetups/modeling-worked-example
https://github.com/neo4j-meetups/modeling-worked-example

